<commit_before>plugins {
    id("java")
    id("jacoco")
    id("maven-publish")
    id("org.jetbrains.kotlin.jvm") version "1.3.21"
}

base.archivesBaseName = "okhttp-mock"

java {
    sourceCompatibility = JavaVersion.VERSION_1_7
    targetCompatibility = JavaVersion.VERSION_1_7
}

dependencies {
    compileOnly("com.squareup.okhttp3:okhttp:3.8.1")
    compileOnly("org.robolectric:robolectric:3.4.2")
    compileOnly("com.android.support:support-annotations:25.3.1")
    compileOnly("com.google.android:android:2.2.1")

    implementation(kotlin("stdlib"))

    testImplementation(configurations.compileOnly)
    testImplementation("junit:junit:4.12")
}

tasks.withType(JacocoReport::class.java) {
    reports {
        xml.isEnabled = true
        html.isEnabled = true
    }

    tasks["check"].dependsOn(this)
}

val sourcesJar by tasks.creating(Jar::class) {
    dependsOn(JavaPlugin.CLASSES_TASK_NAME)
    classifier = "sources"
    from(sourceSets["main"].allSource)
}

val javadocJar by tasks.creating(Jar::class) {
    dependsOn(JavaPlugin.JAVADOC_TASK_NAME)
    classifier = "javadoc"
    from(sourceSets["main"].allSource)
}

publishing {
    publications {
        create<MavenPublication>("default") {
            artifactId = base.archivesBaseName
            from(components["java"])
            artifact(sourcesJar)
            artifact(javadocJar)
        }
    }
    repositories {
        maven {
            name = "bintray"
            url = uri("https://api.bintray.com/content/${System.getenv("BINTRAY_USER")}/maven/okhttp-client-mock/$version")
            credentials {
                username = System.getenv("BINTRAY_USER")
                password = System.getenv("BINTRAY_KEY")
            }
        }
    }
}
<commit_msg>Fixed Bintray publication<commit_after>plugins {
    id("java")
    id("jacoco")
    id("maven-publish")
    id("org.jetbrains.kotlin.jvm") version "1.3.21"
    id("com.jfrog.bintray") version "1.8.4"
}

base.archivesBaseName = "okhttp-mock"

java {
    sourceCompatibility = JavaVersion.VERSION_1_7
    targetCompatibility = JavaVersion.VERSION_1_7
}

dependencies {
    compileOnly("com.squareup.okhttp3:okhttp:3.8.1")
    compileOnly("org.robolectric:robolectric:3.4.2")
    compileOnly("com.android.support:support-annotations:25.3.1")
    compileOnly("com.google.android:android:2.2.1")

    implementation(kotlin("stdlib"))

    testImplementation(configurations.compileOnly)
    testImplementation("junit:junit:4.12")
}

tasks.withType(JacocoReport::class.java) {
    reports {
        xml.isEnabled = true
        html.isEnabled = true
    }

    tasks["check"].dependsOn(this)
}

val sourcesJar by tasks.creating(Jar::class) {
    dependsOn(JavaPlugin.CLASSES_TASK_NAME)
    classifier = "sources"
    from(sourceSets["main"].allSource)
}

val javadocJar by tasks.creating(Jar::class) {
    dependsOn(JavaPlugin.JAVADOC_TASK_NAME)
    classifier = "javadoc"
    from(sourceSets["main"].allSource)
}

val repoDesc = "A simple OKHttp client mock, using a programmable request interceptor Edit"
val repoName = "okhttp-client-mock"
val repoUrl = "https://github.com/gmazzo/$repoName"
val repoTags = arrayOf("okhttp", "mock", "retrofit", "test")

publishing {
    publications {
        create<MavenPublication>("default") {
            artifactId = base.archivesBaseName
            from(components["java"])
            artifact(sourcesJar)
            artifact(javadocJar)
        }
    }
}

bintray {
    user = System.getenv("BINTRAY_USER")
    key = System.getenv("BINTRAY_KEY")
    pkg = PackageConfig().apply {
        repo = "maven"
        name = repoName
        desc = repoDesc
        vcsUrl = "${repoUrl}.git"
        githubRepo = repoUrl.replace("https://github.com/", "")
        githubReleaseNotesFile = "README.md"
        publicDownloadNumbers = true
        setLicenses("MIT")
        setLabels(*repoTags)
    }

    setPublications("default")
}

{'fraction_non_alphanumeric': 0.13197452229299364, 'fraction_numerical': 0.013503184713375796, 'mean_word_length': 3.1457233368532207, 'pattern_counts': {'":': 0, '<': 5, '<?xml version=': 0, '>': 5, 'https://': 3, 'lorem ipsum': 0, 'www.': 0, 'xml': 2}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 1, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '2232167', 'n_tokens_mistral': 1343, 'n_tokens_neox': 1286, 'n_words': 171}